

FreeFEM3D Documentation.

Stéphane DEL PINO, Olivier PIRONNEAU
<http://www.freefem.org>

Version 1.0pre3 (April 16, 2004)

Contents

Contents	iii
1 Generalities	3
1.1 Configurations	3
1.2 Contacts	4
1.2.1 FreeFEM3D's team	4
1.2.2 Project pages	4
1.3 Requirements and Installation	5
1.3.1 Requirements	5
1.3.2 Getting the sources	5
1.3.3 Building the code	6
1.3.4 Getting pre-compiled binaries	8
1.4 Legal conditions	8
1.4.1 Warning	8
1.4.2 GNU General Public Licence	9
2 An overview of FreeFEM3D.	11
2.1 The Problem	11
2.2 The Program using PDEs	11
2.3 Describing the program step by step.	12
2.4 Running the Program	15
3 FEM and EFDM	17
3.1 Finite Element Method.	17
3.2 Embedding of a Fictitious Domain Method.	18
3.2.1 A first approach	18
3.2.2 A second approach	18
3.2.3 Generalities	19
4 Solving problems with FreeFEM3D.	21
4.1 Geometry definition.	21
4.1.1 POV-Ray Language Conventions for FreeFEM3D	22
4.1.2 POV-Ray language basics.	22
4.2 Boolean operations.	23
4.3 Language Basics	24
4.3.1 Simple types.	25
4.3.2 Complex types.	26
4.4 Instructions for the Geometry	28

4.5	Problem definition.	29
4.5.1	The solver bloc.	29
4.5.2	PDE system syntax.	35
4.5.3	Boundary Conditions	36
4.5.4	Bilinear forms.	37
4.5.5	Linear forms.	38
4.5.6	convection operator.	38
4.6	Other Instructions.	39
4.6.1	Input and output.	39
4.6.2	Statements.	40
5	Examples	43
5.1	A Simple Example: the Poisson Problem in a Cube.	43
	Index	45

Acknowledgements

We want here to address special thanks to Robert LI and Alban PAGES for there interest, remarks and encouragements.

Chapter 1

Generalities

FreeFEM3D (aka ff3d) stands for “*FREE Finite Element Method in 3 Dimensions*”.

This software will assist you in solving problems which are modeled by partial differential equations. As the name indicates it is free, (subject to the GPL guidelines), it uses FEM (the finite element method) and it is for three dimensional problems.

You will need to know some mathematics to use FreeFEM3D, because you are required to input the partial differential equations which describe your problem.

Based on our experience with FreeFEM 3.4, FreeFEM+, FreeFEM++ we believe that the best way to describe a problem is through a language adapted to partial differential equations (PDE). Thus for each problem one needs to write a program and submit it to FreeFEM3D which will compile it and run it, and/or report bugs. Therefore it is impossible to use FreeFEM3D without reading some part of this manual or going through some of the examples.

There are 3 steps to solve a PDE

1. Input the geometry and the coefficients
2. build and solve the linear or non-linear discrete systems
3. display graphically the output.

The input of the geometry for a tri-dimensional problem is a formidable task; the entire CAD industry is busy with it. Realizing this FreeFEM3D relies on another program to define the geometry: POV-Ray.

POV-Ray is an image synthesis software which is also free and also runs on a number of operating systems. You will need to learn to use POV-Ray to use FreeFEM3D.

Finally 3D graphics to display the solution of PDEs is also a formidable problem and so FreeFEM3D produces output files which can be visualized with Medit (a simple package written by Pascal FREY) and OpenDx (data explorer), IBM's display software. While Medit runs in MS-Windows, MacOS and Unix (so long as OpenGL is installed), OpenDx is really a unix package and so to run it in windows one needs to install an X11 package (such as Xfree86) and cygwin.

1.1 Configurations

This software requires 64 Megabytes of RAM at least and runs on the following machines:

- Macintosh with MacOS X 10.1 or later;
- PC compatible with MS-Windows (any version using cygwin);

- and GNU/Linux (and others Unix systems)

Note that *compilation* of the program requires nearly 500 Megabytes of memory.

1.2 Contacts

There are several ways for contacting FreeFEM3D's team. The first is to join directly one of its member, but it should be preferred to use the appropriated mailing lists and report bugs using the BTS — all of those features being hosted by the Savannah project.

1.2.1 FreeFEM3D's team

FreeFEM3D's team is composed of the following members:

- Project Director:
Olivier Pironneau <mailto:Olivier.Pironneau@math.jussieu.fr>,
- Main author:
Stéphane Del Pino <mailto:Stephane.DelPino@math.jussieu.fr>,
- Mesh improvements:
Cécile Dobrzynski <mailto:dobrzynski@ann.jussieu.fr>,
- Contributor:
Pascal Havé <mailto:Pascal.Have@math.jussieu.fr>,
- Contributor and Debian Packager:
Christophe Prud'homme <mailto:prudhomme@debian.org>.

1.2.2 Project pages

Being a member of the FreeFEM softwares family FreeFEM3D was developed at the *Laboratoire Jacques-Louis Lions* of the University of Paris VI. FreeFEM3D is hosted at the FreeFEM head quarter: <http://www.freefem.org>. The link that can be used to access directly to the project page is <http://www.freefem.org/ff3d>.

Since FreeFEM3D is a *free* software it is developed transparently: every one can access its source code using cvs. We have chosen to use Savannah as a project development tool. Savannah provides to hosted free softwares:

- code source archiving using cvs,
- mailing lists management,
- Bug Tracking System (BTS),
- and a lot more...

FreeFEM3D's Savannah page is <http://savannah.nongnu.org/projects/ff3d>, any relevant information to use Savannah and FreeFEM3D conjointly will be found there.

Mailing-lists

Four mailing-lists are hosted by Savannah:

- `ff3d-users`: for FreeFEM3D's usage related questions or suggestions;
- `ff3d-dev`: for developers discussions. BTS messages are copied there;
- `ff3d-cvs`: a read-only list that logs cvs messages to inform developers of what changes in the sources, and
- `ff3d-announce`: a read-only low traffic list, used to announce FreeFEM3D's related events...

It is recommended to subscribe to `ff3d-users` and `ff3d-announce`.

The Bug Tracking System

The BTS is the best place to report bugs or wishes. Using it, developers will keep a trace and poster will be automatically informed of any change related to his request, or see what priority has been assigned to it...

1.3 Requirements and Installation

1.3.1 Requirements

If you only want to use a pre-compiled version of FreeFEM3D, you will only have to install the pre-processing and post-processing tools, which are optionally:

- the POV-Ray package (<http://www.povray.org>) which will help you in preparing your geometry,
- OpenDx that is a very powerful and opensource visualization package. Its main drawback may be its complexity — to fix ideas, it uses the same pipeline approach as AVS (check full information at <http://www.opendx.org>).
- An alternative visualization software is `medit`. It is free to download and to use. It was written by P. FREY and can be get at:

<http://www-rocq1.inria.fr/gamma/medit/medit.html>.

1.3.2 Getting the sources

Since FreeFEM3D is developed under the terms of the General Public Licence, it is possible to download, modify and even redistribute its sources¹.

There are two ways of downloading the sources.

¹Note that **binary-only** distributions are forbidden!

Archive files

The first way consists in getting an archive file from the web site at

<http://www.freefem.org/ff3d/sources.html>.

Both zip-like and tar.gz-like archives are provided. Their contents are strictly the same!

After you unpacked the downloaded archive, a new directory called FreeFEM3D, containing the sources will be created.

Using the cvs repository

This second way is probably the best if you want to recompile your own version. Cvs allows you to keep your source code version up to date, it means that after any bug-fix version you just download automatically the modifications. It also provides the possibility to retrieve old versions of the code just specifying a date. Moreover it is possible to download the development version of the code.

In what follows, we will only describe the Unix-like procedure, users of WinCVS should adapt easily it.

Before giving minimalist hints that will help you to download the cvs source tree, we have to insist on some special Savannah's configuration. To improve security Savannah's hackers have chosen to allow only SSHv2 access to their servers. You have to ensure that your CVS_RSH variable is set to ssh. Using a Bourn-like shell (sh, ksh, bash, zsh...) use the instruction:

```
export CVS_RSH=ssh
```

If you run a C-like shell:

```
setenv CVS_RSH ssh
```

Since you will need this to be set *before* executing *each* cvs command, it is recommended that you set it once for all in the appropriate shell start-up file. Reading cvs documentation may be a useful help: check its web site at <http://www.cvshome.org> to get more information.

Let us now recall the basic cvs commands that will allow you to maintain your own cvs-tree synchronized.

Checking-out the code is required only once. You will have to do it to get your first copy of the sources. Executing the instruction line:

```
cvs -d :ext:anoncvs@savannah.gnu.org:/cvsroot/ff3d co ff3d
```

will create a ff3d directory containing the current development version of the package.

Keeping FreeFEM3D up to date will only require the command:

```
cvs -z3 update
```

See <http://savannah.nongnu.org/cvs/?group=ff3d> for more details.

1.3.3 Building the code

FreeFEM3D needs several tools to be built. Some of them are optional, others are just essential. Note that the code compilation requires lots of memory when building an optimized version.

All of the following softwares are common Unix packages. They should also be found on MacOS X and are provided by cygwin when building for MS-Windows.

Optional packages

Up to now, only two packages are optional: Autogen and VTK.

`Autogen` is a tool designed for generating programs. It is used in FreeFEM3D to generate a fancy line parser as well as Unix man and info pages. If you do not use it for compilation, FreeFEM3D will not accept any options, the only argument will be the FreeFEM3D command file. Autogen is part of the GNU project and can be downloaded from its web site.

VTK stands for “the Visualization Toolkit”, it is an opensource C++ library developed by KitWare that provides high level facilities to perform scientific graphics in 2D or 3D. Its presence in FreeFEM3D is still experimental and undocumented. To get this library, connect to <http://www.vtk.org>.

Required packages

Only few packages are required to compile FreeFEM3D. They are essentially compilers or building tools.

`Bison` is the GNU implementation of yacc which stands for *yet another compiler compiler*. This software is dedicated to the construction of languages parsers. Since FreeFEM3D uses two languages (one for the problem description, the other for the geometry) it uses such a tool to generate their compilers. Being a part of the GNU project, bison can be downloaded from <http://www.gnu.org>.

`Automake/Autoconf` are used to generate the Makefiles. They are in charge of the package configuration and build dependencies. Each of the packages of this family have to be installed and particularly `libtool`.

An ANSI C++ compiler is also required. FreeFEM3D use some bleeding edge C++ constructions that need good compiler. Recent GNU GCC compilers offer very good ANSI C++ implementation. FreeFEM3D has been developed using those tools. We recommend the use of g++ version 3.2 to build the sources.

`make` is obviously needed to build FreeFEM3D. The GNU version, sometimes called `gmake` is recommended as well.

Building instructions

FreeFEM3D uses a `configure` script to detect your configuration and generate Makefiles. If you downloaded an archive, this script should be found in the `ff3d` directory. If it misses or if you used `cvs` to get the code, you have to generate it. This is very simple. Enter the `ff3d` directory — all the following commands will be performed from this particular place. Now, type the command

```
autoreconf -i
```

This can produce warnings saying that some files are replaced, it is normal. You can now call the `configure` script.

The `configure` script uses the standard configure options. To get the complete list of them, type

```
./configure --help
```

at the shell prompt. We will now discuss the FreeFEM3D special options. Note that most of “--enable-” option have an opposite “--disable-” option.

`--enable-debug` is used to build a debugging version which runs slower but provide. A standard optimized version is generated if this option is omitted.

`--enable-opt` generates an even more optimized version (using none standard g++ options). This option conflicts with the `--enable-debug` option.

`--enable-vtk` allows the generation of the code with VTK support. This option is automatically enabled if VTK is detected — but can still be deactivated using the `--disable-vtk` option

`--enable-exec` permits the use of the `exec` instruction in FreeFEM3D files. It is enabled by default. Since it allows execution of external programs within FreeFEM3D, it is a potentially dangerous for security. This is why it can be deactivated.

`--enable-real_t` is used to change the type of variable to store reals. Default value is `double`, this can be changed to `float` using `--enable-real_t=float`. Others types should be added in the future.

The execution of the `configure` script, creates the Makefiles. Typing

```
make
```

will generate the executable called `ff3d` or `ff3d.exe`, for the MS-Windows version, after a few² compilation time.

1.3.4 Getting pre-compiled binaries

Binary files for common architectures and systems can be downloaded at

<http://www.freefem.org/ff3d/binaries.html>.

If you are the happy owner of a Debian GNU/Linux³ system, you can install it by the simple command:

```
apt-get install freefem3d
```

1.4 Legal conditions

1.4.1 Warning

FreeFEM3D is a scientific product to help you solve Partial Differential Equations in 3 dimensions; it assumes a basic knowledge and understanding of the Finite Element Method and of the Operating System used. It is also necessary to read carefully this documentation to understand the possibilities and limitations of this product. The authors are **not responsible** for any errors or damages due to wrong results.

²or a bit more ;-))

³At the time of writing these lines the `freefem3d` package is only available in the `unstable` release.

1.4.2 GNU General Public Licence

It is in its name: FreeFEM3D is a *free software*. It is distributed under the GNU GPL guidelines as said here:

This program is free software; you can redistribute it and/or modify it under the terms of the GNU General Public License as published by the Free Software Foundation; either version 2, or (at your option) any later version.

This program is distributed in the hope that it will be useful, but WITHOUT ANY WARRANTY; without even the implied warranty of MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. See the GNU General Public License for more details.

You should have received a copy of the GNU General Public License along with this program; if not, write to the Free Software Foundation, Inc., 59 Temple Place - Suite 330, Boston, MA 02111-1307, USA.

See the COPYING file in ff3d's root directory, the exact terms of this licence can also be consult online at GNU project official web site: <http://www.gnu.org/copyleft/gpl.html>.

Chapter 2

An overview of FreeFEM3D.

In this chapter we will focus on a fairly complex example so that the user will learn the existence of the main possibilities of FreeFEM3D. In the next chapters we will use simpler examples while focusing on different features. Finally in the last chapter more complex examples will be proposed.

2.1 The Problem

Consider the following coupled problem: find (u, v) such that

$$\begin{cases} -\Delta u + v = f \text{ in } \mathcal{O}, \\ u = x(x-1) + y(y-1) + z(z-1) \text{ on } \partial\mathcal{O}, \\ v - \Delta v + u = g \text{ in } \mathcal{O}, \\ v = \sin(\pi x) \sin(\pi y) \sin(\pi z) \text{ on } \partial\mathcal{O}. \end{cases} \quad (2.1)$$

where $f = -6 + \sin(\pi x) \sin(\pi y) \sin(\pi z)$,
and $g = (3\pi^2 + 1)f + x(x-1) + y(y-1) + z(z-1)$.

which has an analytical solution:

$$(u, v) = (x(x-1) + y(y-1) + z(z-1), \sin(\pi x) \sin(\pi y) \sin(\pi z))$$

whatever the domain \mathcal{O} and its boundary $\partial\mathcal{O}$ are.

2.2 The Program using PDEs

To solve (2.1) in the domain \mathcal{O} that is the sphere centered in 0 and whose radius is 0.8 we can write the following program (example1.txt).

```
example1.txt
1 vector n = (20, 20, 20);
2 vertex a = (-1,-1,-1);
3 vertex b = ( 1, 1, 1);
4 double pi = 4*atan(1);
5
```

```

6 scene Sc = pov("example1.pov"); // POV-Ray geometry file
7
8 mesh M = structured(n,a,b);
9
10 domain O = domain(Sc,inside(<1,0,0>));
11
12 function Ki0 = one(<1,0,0>);
13
14 function uexact = (x*(x-1) + y*(y-1) + z*(z-1));
15 function vexact = sin(pi*x)*sin(pi*y)*sin(pi*z);
16
17 solve(u,v) in O by M method(type=penalty)
18 {
19     pde(u)
20         - dx(dx(u)) - dy(dy(u)) - dz(dz(u)) + v
21         = -6 + vexact;
22     u = uexact on <1,0,0>;
23
24     pde(v)
25         v - div(grad(v)) + u
26         = (3*pi^2 + 1)*vexact + uexact;
27     v = vexact on <1,0,0>;
28 }
29
30 double I=int(M)(Ki0*(uexact-u)^2);
31 double J=int(M)(Ki0*uexact^2);
32 cout << sqrt(I/J) << "\n";
33 I = int(M)(Ki0*(vexact-v)^2);
34 J = int(M)(Ki0*vexact^2);
35 cout << sqrt(I/J) << "\n";
36
37 save(opendx,"v.dat",Ki0*v,M);
38 save(opendx,"u.dat",Ki0*u,M);

```

The geometry is given in a POV-Ray file (example1.pov)

```

1 sphere {
2     <0,0,0>, 0.8
3     pigment { color rgb <1,0,0> }
4 }

```

2.3 Describing the program step by step.

From this example we see that there are different *types* such as

vector, vertex, scene, mesh, domain, double, function

and that each instruction is ended using a *semicolon*. The syntax is borrowed from the C/C++ language, whenever possible.

Lets go through the example:

```

1  vector n = (20, 20, 20);
2  vertex a = (-1,-1,-1);
3  vertex b = ( 1, 1, 1);
4  double pi = 4*atan(1);

```

Those four lines define n , a and b as \mathbb{R}^3 elements and the *double precision number* π as π . These will be used later to define a box from the two points a and b meshed by an (n_x, n_y, n_z) uniform finite difference grid.

```

5
6  scene Sc = pov("example1.pov"); // POV-Ray geometry file

```

The *scene* Sc is defined using the POV-Ray description contained in the file `example1.pov`. Since no path was given here, the file must be in the current directory, the same that contains `example1.txt`. Note also that C^{++} -like commentaries can be used.

```

7
8  mesh M = structured(n,a,b);

```

Here we construct the structured mesh M using $20 \times 20 \times 20$ vertices in each direction (this is given by the value of n). Vectors a and b are to be two vertices of the same diagonal defining a box. The frame (x, y, z) is *direct* so if (Ox) is horizontal from left to right and (Oy) is vertical bottom to top on the screen then (Oz) points towards you. This is the *right hand rule*.

```

9
10 domain O = domain(Sc,inside(<1,0,0>));

```

The computational *domain* O is declared for future use. Its definition uses the objects of the *scene* Sc (Domain declaration can take more arguments; see below). When the only argument is a *scene*, it means that the computation domain can be

- the entire *box* defined by the structured mesh M (it is then a standard *finite element* approximation), or
- a more complex domain $O \subset M$, then the *fictitious domain method* has to be implemented by the *user* and FreeFEM3D provides tools for this (see 3.2).

This is the second case here. For more details about the method see the section 3.

```

11
12 function KiO = one(<1,0,0>);

```

Here we define the *function* KiO which is defined as the *indicator* function of *objects* whose color is $\langle 1, 0, 0 \rangle$ in the POV-Ray file: that:

$$KiO(x, y, z) = \begin{cases} 1 & \text{if } (x, y, z) \text{ is inside at least one object of color } \langle 1, 0, 0 \rangle. \\ 0 & \text{else.} \end{cases}$$

Remark that function allows the declaration of *analytical* functions. *There is no approximation at this point!* KiO is the *exact* function.

```

14 function uexact = (x*(x-1) + y*(y-1) + z*(z-1));
15 function vexact = sin(pi*x)*sin(pi*y)*sin(pi*z);

```

Here we define two more functions: `uexact` and `vexact`:

$$\begin{cases} u_{\text{exact}} = x(x-1) + y(y-1) + z(z-1) \\ v_{\text{exact}} = \sin(\pi x) \sin(\pi y) \sin(\pi z) \end{cases}$$

One can see that FreeFEM3D supports the algebra of functions. `x`, `y` and `z` have to be seen as functions by abuse of notation.

```

17 solve(u,v) in O by M method(type=penalty)
18 {
19   pde(u)
20     - dx(dx(u)) - dy(dy(u)) - dz(dz(u)) + v
21     = -6 + vexact;
22   u = uexact on <1,0,0>;
23
24   pde(v)
25     v - div(grad(v)) + u
26     = (3*pi^2 + 1)*vexact + uexact;
27   v = vexact on <1,0,0>;

```

This instruction bloc defines *both* the PDE problem and how to *solve* it! Lets now focus to the details.

```

16
17 solve(u,v) in O by M method(type=penalty)

```

means that we are going to *solve* a *coupled* PDE problem whose *unknowns* are the *functions* `u` and `v`, defined on the set \mathcal{O} , and that to solve the problem we will use the *mesh* `M`.

`method(type=penalty)` is optional, it is here to say to the solver to use *penalty* method for boundary conditions. Many options can be passed to the solver.

The next block defines the two coupled PDEs. The first

```

19 pde(u)
20   - dx(dx(u)) - dy(dy(u)) - dz(dz(u)) + v
21   = -6 + vexact;

```

defines the PDE on `u`:

$$-\partial_x \partial_x u - \partial_y \partial_y u - \partial_z \partial_z u + v = -6 + v_{\text{exact}} \text{ in } \mathcal{O}.$$

Using this formulation we illustrate the fact that *general second order* operator can be approximated. Will see that non constant coefficient can be used. The computational domain is provided on line 17.

Then,

```

22         u = uexact on <1,0,0>;
23

```

defines the *Dirichlet* boundary conditions associated to the unknown u : $u = u_{\text{exact}}$ on $\partial\mathcal{O}$, where $\partial\mathcal{O}$ is the boundary of objects which *rgb color* is $\langle 1, 0, 0 \rangle$ (red).

Now, the second equation is given, using a more compact form

```

24     pde(v)
25         v - div(grad(v)) + u
26         = (3*pi^2 + 1)*vexact + uexact;
27     v = vexact on <1,0,0>;

```

Which is

$$\begin{cases} v - \nabla \cdot \nabla v + u = (3\pi^2 + 1)v_{\text{exact}} + u_{\text{exact}} \text{ in } \mathcal{O}. \\ \text{with } v = v_{\text{exact}} \text{ on } \partial\mathcal{O}. \end{cases}$$

Remark 1 One should note that boundary condition on each variable are given while describing the associated PDE. The boundary condition process has to be understood precisely (see section 4.5.3). Moreover, in the case of PDE systems users may find variational formulae better suited and less prone to errors.

Finally to check the results against the analytical solution we print the relative L^2 errors

$$\left| \frac{(\int_{\mathcal{M}} \mathbf{1}_{\mathcal{O}}(u - u_{\text{exact}})^2 / \int_{\mathcal{M}} \mathbf{1}_{\mathcal{O}} v_{\text{exact}}^2)^{\frac{1}{2}}}{(\int_{\mathcal{M}} \mathbf{1}_{\mathcal{O}}(v - v_{\text{exact}})^2 / \int_{\mathcal{M}} \mathbf{1}_{\mathcal{O}} v_{\text{exact}}^2)^{\frac{1}{2}}} \right| \text{ and,}$$

computed the following way:

```

30     double I=int(M) (Ki0*(uexact-u)^2);
31     double J=int(M) (Ki0*uexact^2);
32     cout << sqrt(I/J) << "\n";
33     I = int(M) (Ki0*(vexact-v)^2);
34     J = int(M) (Ki0*vexact^2);
35     cout << sqrt(I/J) << "\n";

```

and then save the solution in the files `v.dat` and `u.dat`.

```

37     save(appendx,"v.dat",Ki0*v,M);
38     save(appendx,"u.dat",Ki0*u,M);

```

2.4 Running the Program

There is no graphic interface in FreeFEM3D; only text output. As linear systems are solved iteratively, for each iteration we display the residual error. Running the above program produces the following output:

```

TO FILL

```

Results are shown on figure 2.1.




Figure 2.1: u and v iso-values on the cutting plane passing by $(\frac{1}{2}, \frac{1}{2}, \frac{1}{2})$ parallel to (Ox, Oy) .

Chapter 3

FEM and EFDM

3.1 Finite Element Method.

The Finite Element Method (FEM) is a Galerkin method applied to the variational formulation of the problem. In the case of (2.1) for instance the variational formulation consists in finding $u, v \in H^1(\Omega)$ satisfying the boundary conditions ($u|_\Gamma = u_\Gamma, v|_\Gamma = v_\Gamma$) and

$$\int_{\Omega} (\nabla u \cdot \nabla \hat{u} + v \hat{u} + (v + u) \hat{v} + \nabla v \cdot \nabla \hat{v}) - \int_{\Omega} (f \hat{u} + g \hat{v}) = 0 \quad \forall \hat{u}, \hat{v} \in H_0^1(\Omega), \quad (3.1)$$

where H_0^1 is the subspace of H^1 of functions which are zero on the boundaries and H^1 is the space of square integrable functions with square integrable derivatives.

The Galerkin method consists in approximating the problem by replacing $H^1(\Omega)$ by a finite subspace, a special case is the FEM of order 1 on a tetraedra grid:

$$H_h = \{w_h \text{ continuous, piecewise linear on the tetraedrisation of } \Omega\}. \quad (3.2)$$

Then, the discretized problem can be: find $u_h, v_h \in H_h$ such that $u_h = u_\Gamma$ and $v_h = v_\Gamma$ on Γ and such that:

$$\begin{aligned} \int_{\Omega} (\nabla u_h \cdot \nabla \hat{u}_h + v_h \hat{u}_h + (v_h + u_h) \hat{v}_h + \nabla v_h \cdot \nabla \hat{v}_h) \\ - \int_{\Omega} (f \hat{u}_h + g \hat{v}_h) = 0 \quad \forall \hat{u}_h, \hat{v}_h \in H_h \cap H_0^1(\Omega). \end{aligned} \quad (3.3)$$

Dirichlet boundary conditions can be implemented by *elimination* or by *penalty*; in this later case, the discrete problem is: find $u_h, v_h \in H_h$ such that

$$\begin{aligned} \int_{\Omega} (\nabla u_h \cdot \nabla \hat{u}_h + v_h \hat{u}_h + (v_h + u_h) \hat{v}_h + \nabla v_h \cdot \nabla \hat{v}_h) - \int_{\Omega} (f \hat{u}_h + g \hat{v}_h) \\ + \frac{1}{\epsilon} \int_{\Gamma} (u_h - u_\Gamma) \hat{u}_h + (v_h - v_\Gamma) \hat{v}_h = 0 \quad \forall \hat{u}_h, \hat{v}_h \in H_h \end{aligned} \quad (3.4)$$

where ϵ is a small parameter and Γ denotes the boundary of Ω .

3.2 Embedding of a Fictitious Domain Method.

3.2.1 A first approach

The Embedding of a Fictitious Domain Method (EFDM) tries to avoid the difficulty of dividing Ω into non-overlapping tetraedra and so extends all functions in a simpler domain C containing Ω .

So we denote by

$$V_h = \{w_h \text{ continuous, piecewise linear on the tetraedrisation of } C\} \quad (3.5)$$

and solve for $u_h, v_h \in V_h$:

$$\begin{aligned} \int_C \mathbf{1}_\Omega (\nabla u_h \cdot \nabla \hat{u}_h + v_h \hat{u}_h + (v_h + u_h) \hat{v}_h + \nabla v_h \cdot \nabla \hat{v}_h) - \int_C \mathbf{1}_\Omega (f \hat{u}_h + g \hat{v}_h) \\ + \frac{1}{\epsilon} \int_\Gamma (u_h - u_\Gamma) \hat{u}_h + (v_h - v_\Gamma) \hat{v}_h = 0 \quad \forall \hat{u}_h, \hat{v}_h \in V_h. \end{aligned} \quad (3.6)$$

Remark 2 One should note that (3.6) does not have a unique solution in V_h . This comes from the fact that the operator contained in (3.6) coincides with the $\mathbf{0}$ operator in $C \setminus \Omega$. In other words, (3.6) could be written as $\mathbf{0}(\frac{u}{v}) = 0$ in $C \setminus \Omega$ and (3.4). One cannot use direct method to solve the associated linear system but since we use a Conjugate Gradient-like method the solution of the discrete problem will converge to the solution of the continuous problem in Ω .

The program given in 2.2 uses this approach.

3.2.2 A second approach

Let us look to the simpler problem:

$$\begin{cases} \nabla \cdot \nu \nabla u = f \text{ in } \Omega \\ u|_\Gamma = u_\Gamma, \end{cases} \quad (3.7)$$

where $\Gamma = \partial\Omega$.

Now, we define C such that $\Omega \subset C$ and $\partial\Omega \cap \partial C = \emptyset$. Let $\tilde{u} \in H_0^1(C)$ such that $\tilde{u} = u$ in $H^1(\Omega)$, and let $\tilde{\nu} \in L^\infty(C)$ such that $\tilde{\nu} = \nu$ in $L^\infty(\Omega)$, and $\tilde{\nu}$ is strictly positive in $C \setminus \Omega$.

To simplify notations let's now call \tilde{u} : u and $\tilde{\nu}$: ν .

One can note that penalty amounts to trade the Dirichlet boundary condition for the Fourier condition:

$$\frac{u - u_\Gamma}{\epsilon} + \left[\nu \frac{\partial u}{\partial n} \right] = 0 \quad (3.8)$$

where $[\cdot]$ stands for the jump across Γ .

Neumann and Robin jump conditions such as in the problem

$$\begin{cases} \nabla \cdot (\nu \nabla u) = f \text{ in } C \setminus \Gamma \\ \beta u + \left[\frac{\partial u}{\partial n} \right] = g \text{ on } \Gamma, \end{cases} \quad (3.9)$$

has the variational formulation: find $u \in H_0^1(C)$

$$\int_C (\nu \nabla u \cdot \nabla \hat{u} - f \hat{u}) + \int_\Gamma (\beta u - g) \hat{u} = 0 \quad \forall \hat{u} \in H_0^1(C). \quad (3.10)$$

Therefore Neumann and Robin conditions require an extension of the operator so that $\nu \partial_n u = 0$ on Γ^+ (n^+ being the outer normal). One way is to take $\nu \ll 1$ outside Ω .

3.2.3 Generalities

In FreeFEM3D you may either do the EFDM by yourself or let it be done automatically by specifying the domain (see section ??). There are others approaches to EFDM [?][?][?].

Add others remarks and Robin/Neumann boundary conditions description?

Chapter 4

Solving problems with FreeFEM3D.

4.1 Geometry definition.

A PDE problem is defined by a set of PDEs and a computation domain. As it is described in section 3 FEM uses a tetraedrisation of the domain. To build the tetraedrization of the domain is a complex task that we avoid here by using EFDM.

EFDM will require to define functions which take different values inside and outside objects and also to compute boundary integrals on the objects.

In FreeFEM3D, the geometry of the domain is given using *Virtual Reality* (VR) data. It means that the domain is defined as *set operations*¹ on simple *primitive shapes*². For various technical reasons, the language chosen to describe VR is POV-Ray's.

Image synthesis softwares such as POV-Ray define *scenes* as a collection of simple objects with set operations on them. But they also worry about realistic rendering and so a number of features are irrelevant for us, such as the camera, the type of light, the textures³. What we need is simply to define a scene as a collection of objects, know what are the set operations that have been applied to them and name each object. Therefore, for a scene made of a sphere and a brick, the following is sufficient:

```
sphere {  
  <0,0,0>, 1.5  
  pigment { color rgb <0,1,0> }  
}  
box {  
  <0,0,0>,  
  <2,2,2>  
  pigment { color rgb <0,0,1> }  
}
```

In addition it may be helpful to visualize the fictitious computational domain. So the scene may also contain (but only for the visualization) the computational box, for example

```
box {  
  <-2,-2,-2>,
```

¹set operations are union, extrusion and intersection.

²box, sphere, cylinder, cone,...

³One should note that those keywords are just **ignored** by FreeFEM3D compiler, so you don't have to modify the scene you rendered with POV-Ray.

```

    < 3, 3, 3>
    pigment { color rgbf <1,1,1,0.5> }
}

```

The color which is assigned to each object will be used by *FreeFEM3D* to identify the object and its boundary condition, so it is important to distinguish colors if boundary conditions are different.

Colors like `rgb` `<a,b,c,d>` define objects which will not be identified by *FreeFEM3D*, and so are used only for graphics.




Figure 4.1: A green sphere and a blue cube. The transparent box is for graphic use only (to see the computational domain used) but *FreeFEM3D* will check that all objects are contained in it.

The user is sent to the POV-Ray language manual for a complete reference, but here are some *conventions* used in *FreeFEM3D* and some basis to the POV-Ray language.

4.1.1 POV-Ray Language Conventions for *FreeFEM3D*

References: In *Scientific Computing* it is common practice to define physical parameters and boundary conditions using *reference* (indices that will characterize *degrees of freedom*).

Since the geometry description is not contained in the mesh, but comes from a POV-Ray file, *FreeFEM3D* uses *objects colors* as references. So the reference is not given by an *integer* but by an \mathbb{R}^3 vector such as:

```

<a,b,c>

```

where a , b and c are three *double*. The colors in POV-Ray being given by values of red, green and blue. So, if one wishes to visualize the scene before computation he must choose a , b and c in $[0, 1]$; if not, values are not bounded...

4.1.2 POV-Ray language basics.

sphere:

```

sphere {
    <0,0,0>, 1
    pigment { color rgb <1,0,0> }
}

```

defines a *unit* sphere centered at $(0,0,0)$. The pigment is defined so that the sphere will be referenced as $\langle 1,0,0 \rangle$.

box:

```
box {  
    <0,0,0>, <1,1,1>  
    translate <1,-1,1>  
    pigment { color rgb <1,0,1> }  
}
```

describes a box built on vertices $(0,0,0)$ and $(1,1,1)$ and *then* translate by a vector $(1,-1,1)$. It is referenced has $\langle 1,0,1 \rangle$.

cylinder:

```
cylinder {  
    <0,0,0>, <1,1,1>, 0.2  
    pigment { color rgb <1,0,1> }  
}
```

is a cylinder built on the axis defined by $(0,0,0)$ and $(1,1,1)$ whose radius is 0.2. Its reference is $\langle 1,0,1 \rangle$.

4.2 Boolean operations.

The main interest of CSG is the ability of combining all primitives and built objects. POV-Ray's way of doing it is given bellow.

object:

```
object {  
    box {  
        <0,0,0>, <1,1,1>  
    }  
    pigment { color rgb <0,0,1> }  
}
```

Defines a box and *then* using the object statement the reference $(0,0,1)$ is assigned to it.

union:

```
union {  
    box {  
        <0,0,0>, <1,1,1>  
    }  
    box {
```

```

    <0,0,0>, <1,1,1>
    translate <1,-1,1>
  }
  pigment { color rgb <1,0,1> }
}

```

defines the union of two boxes. The second is translated by a vector $(1, -1, 1)$. The obtain object has the reference $(1, 0, 1)$. The union can be operated on n objects.

Remark 3 *One can substitute the keyword `merge` to `union`. There is strictly no difference in FreeFEM3D as opposed to POV-Ray.*

intersection:

```

intersection {
  cylinder {
    <0,0,0>, <1,1,1>, 1
  }
  box {
    <0,0,0>, <1,1,1>
  }
  pigment { color rgb <1,0,1> }
}

```

builds the intersection of a cylinder and a box. The obtain object has the reference $(1, 0, 1)$. The intersection can be operated on n objects.

difference:

```

difference {
  sphere {
    <0,0,0>, 1
  }
  box {
    <0,0,0>, <1,1,1>
  }
  pigment { color rgb <1,0,1> }
}

```

extrudes a box from a sphere. The obtain object has the reference $(1, 0, 1)$. The difference can be operated on n objects, then, the $n - 1$ objects are extruded from the first one.

4.3 Language Basics

FreeFEM3D tries to look like *C++* in a reasonable way! So, people familiar to *C* or *C++* should learn the language easily (the biggest difficulty being that one does not know which subset of

C++ is implemented). At the same time, someone unfamiliar with those languages should not be afraid since it is a *hi-level*⁴ language.

Knowing that we will explain step-by-step this language, let us first recall important rules:

- Every variable must be declared before being used.
- One cannot declare a variable twice.
- Each variable is global, so when declared, it *lives* until the end of the program execution, even if declared *in a block*. This behaviour should change in the future.

4.3.1 Simple types.

In the following we describe the two basis types of variables that can be declared in FreeFEM3D: `double` and `vector/vertex` ($\in \mathbb{R}^3$). The language supports other types such as *boolean* or *string* but they are only used internally (up to now).

Variable declaration follows the general syntaxe

```
<typeid> <variableid>;
```

which *declares* a *non initialized variable*, or

```
<typeid> <variableid> = <typevalue>;
```

which *constructs* a variable initialized to the given value.

double: doubles are used to represent \mathbb{R} elements. To declare a double, use the following syntax:

```
double a = 3.14159;
```

The *algebra* on \mathbb{R} has been implemented, and classical function are built in the language, so one can write

```
1 double pi = 4*acos(1);
2 double b;
3 b = sqrt(1+pi^2*(2+sqrt(2)));
```

In this case, `b` will contain, as expected, the value $1 + \pi^2(2 + \sqrt{2})$ at line 3, before (line 2) its content is not determined.

The table 4.1 shows FreeFEM3D functions that can be used on `double`.

⁴*hi-level* in the sense that it manipulates “real-life” objects.

FreeFEM3D syntax	mathematical meaning
<code>abs(a)</code>	$ a $
<code>exp(a)</code>	e^a
<code>log(a)</code>	$\log(a)$
<code>sin(a)</code>	$\sin(a)$
<code>cos(a)</code>	$\cos(a)$
<code>tan(a)</code>	$\tan(a)$
<code>asin(a)</code>	$\arcsin(a)$
<code>acos(a)</code>	$\arccos(a)$
<code>atan(a)</code>	$\arctan(a)$
<code>a^b</code>	a^b

Table 4.1: Built-in functions on doubles. a and b are double.

vector/vertex: To declare a vector one can use one of the two following syntax:

```
vector v1 = (1,0,0);
vertex a = (0,1,0);
```

By now, there is no difference between `vertex` and `vector` but the `vertex` object could be enriched for future use.

Also, in a sense of POV-Ray compatibility, one can use indifferently the notations (a, b, c) or $\langle a, b, c \rangle$. We suggest the use of $\langle a, b, c \rangle$ only when the vector refers to the color of a POV-Ray object.

Since \mathbb{R}^3 algebra is implemented, one can write the following expression:

```
vector w = 2*(1,1,1)+(-2*2.5,1,2);
```

then w will contain $(-3, 3, 4)$.

4.3.2 Complex types.

By complex types we do not mean “composed” types such as structures or classes but types whose behavior needs some enlightenment.

function: Scalar functions algebra is implemented in FreeFEM3D. It means that the user can manipulate scalar functions in an analytic way, *i.e.* manipulating the functions and not their approximations.

There are 3 special functions/keywords: x, y, z which are $\mathbb{R}^3 \rightarrow \mathbb{R}$ functions which refer to the coordinate system:

$$(x, y, z) \mapsto x, \quad (x, y, z) \mapsto y, \quad (x, y, z) \mapsto z. \quad (4.1)$$

One can now easily define polynomial functions:

```
function p = 2*x + y + x*z + z^3;
```

with the obvious meaning $p(x, y, z) = 2x + y + xy + z^3$.

The built-in functions in FreeFEM3D are listed in table 4.2. For example, one could redefine the $(x, y, z) \mapsto \tan(x)$ function using the name tag `tan_x` by:

```
function tan_x = sin(x)/cos(x);
```

It is possible define functions with *constant* values by assigning a double to the function, such as in

```
function cos1 = cos(1);
```

FreeFEM3D syntax	mathematical meaning
<code>abs(f)</code>	$ f $
<code>exp(f)</code>	e^f
<code>log(f)</code>	$\log(f)$
<code>sin(f)</code>	$\sin(f)$
<code>cos(f)</code>	$\cos(f)$
<code>tan(f)</code>	$\tan(f)$
<code>asin(f)</code>	$\arcsin(f)$
<code>acos(f)</code>	$\arccos(f)$
<code>atan(f)</code>	$\arctan(f)$
<code>f^g</code>	f^g

Table 4.2: Built-in functions on functions. f and g are functions.

femfunction: This is a special type related to finite element functions (see appendix ?? for more details). Up to now, only Q^1 functions are allowed: piece-wise *tri-linear*⁵ and continuous functions defined on an hexahedral mesh. Since finite element functions spaces needs a *mesh* to be defined, one has to write

```
femfunction f(M) = sin(x);
```

where M is a previously defined mesh (see 4.3). After the instruction, f contains a finite element approximation of $(x_1, x_2, x_3) \mapsto \sin(x_1)$. f is in fact defined by the interpolation at mesh vertices: $f(X) = \sin(X_1)$ if X is a vertex of M .

Even though a femfunction is a special kind of function, it can be used in the function algebra:

```
femfunction f(M) = exp(x+y*z);
function g = x - f;
```

Again, this is a *definition*; g is not evaluated at this point. The evaluation will be performed when needed as in

```
double t = g(1,1,1);
```

which not only defines t but also triggers an evaluation of g at $(1,1,1)$, and therefore the computation of the linear interpolation to calculate $f(1, 1, 1)$.

⁵tri-affine for purists!

4.4 Instructions for the Geometry

scene: VR data comes from a POV-Ray file. The geometry informations contained in this file are stored in a scene variable.

The syntax is very simple:

```
scene S = pov("scene.pov");
```

Then S contains the POV-Ray *scene* described in the file `scene.pov`.

One can define many *scenes* in one FreeFEM3D file. In that latter case, to avoid ambiguities, the last *used* scene is the current one. Lets look at the following example:

```
1 scene S1 = pov("scene1.pov");
2 scene S2 = pov("scene2.pov");
3 function f = one(<1,0,0>);
4 using S1;
5 f = one(<1,0,0>);
```

After line 2, two scenes S1 and S2 are defined and *current scene is* S2. So at line 3, the function `f`, is the *indicator* function of the object `<1,0,0>` of *scene S2*.

Line 4, the *current scene* is set to S1! Line 5 makes the function `f` be the *indicator* function of the object `<1,0,0>` of *scene S1*.

domain: The computational *domain* is defined using a scene,

```
domain <domain> [ = domain(<scene> , <booleanexp>) ];
```

where the `booleanexp` is defined by the following:

```
<booleanexp>: {not <inoutexp> | <inoutexp>}
```

and finally

```
<inoutexp>: { ( <inoutexp> )
              | <inoutexp> and <inoutexp>
              | <inoutexp> or <inoutexp>
              | outside(<ref>)
              | inside(<ref>)} 
```

Remark 4 The keywords *not*, *and* and *or* can respectively be replaced by `!`, `&&` and `||`, as it is the case for every boolean operation.

mesh: There are two kinds of meshes in FreeFEM3D: volume and surface. The volume meshes are only *cartesian structured* meshes, *i.e.* a squared box regularly meshed by squared hexahedra. Surface meshes are composed of triangles or quadrangles living in \mathbb{R}^3 , they are only used to compute integrals in the fictitious domain method. Here is how to deal with them.


```
vector n = (10,10,10);
vector a = (0,0,0);
vector b = (1,1,1);
mesh M = structured(n,a,b);
```

With this code, M will be a *uniform* grid of the box $]0,1[^3$ using 10 vertices in each direction. To construct a *surface* mesh, one needs a *structured* mesh and can write

```
mesh m = surface(<1,0,0>, S, M);
```

where $\langle 1,0,0 \rangle$ is the *reference* of a POV-Ray object, given by the scene S and M a *structured* mesh; all of them having previously been declared.

Remark 5 *The surface mesh is built using a marching cubes-like method, this is why a structured mesh is needed. In future, reading meshes in files should be allowed, also.*

4.5 Problem definition.

4.5.1 The solver bloc.

solve: The *solve block* is of course special to FreeFEM3D and should be studied with attention:

```
solve (<unknown list>) in <domain> by <mesh> [<solver options>]
{
  { pde (<unknown1>)
    <pde>
    <boundary condition list>
  [ pde(< unknown2>)
    <pde>
    <boundary condition list>
    [...]]
  ]
  | test(<test function list>)
    <variational formula>
    <dirichlet boundary condition list> }
}
```

Remark 6 *Note that the solve bloc can be used with two different kinds of body. The first one is systems of PDEs, the second is variational formulae.*

Lets focus on the common part and then on the $\langle \text{unknown list} \rangle$, it is a set of unknown of the form:

```
u, v, w
```

Those unknowns can already have been defined has functions! If it is the case, there value will be the first guess for iterative methods, otherwise it will be 0. An instruction like

```
in <domain> by <mesh>
```

gives informations to the solver as to which <domain> and which <mesh> use.

Remark 7 If the *unknown* list contains n elements, there must be n *pde* statements (one per *unknown*)

Remark 8 The *unknowns* computed by the solver can be used later as if they were declared as *femfunctions* of the solving mesh (whatever they were before).

Remark 9 If an *unknown* was previously defined as a function its interpolate will be taken as the initial guess for iterative methods.

Solver options The <solver options> is used to pass arguments to the solver such as the type of algorithm to solve the linear system (conjugate gradient, bi-conjugate gradient) and associated parameters (type of preconditioner, maximum number of iterations, value of ϵ), discretization method,...

Below is a list of solver options⁶

The syntax of options is

```
[ <option name> ( <suboption> = <value> [,
                  <suboption> = <value> ]...) [,
...]]
```

Here comes the complete list of options.

FreeFEM3D treats the options the following way: the parser reads the option set and builds a tree associated to it. Then when solve starts, each *parametrizable* object reads its options when it is built. It looks first in the tree and if an option is not specified here, it uses the default value.

Knowing the rules, let's look at some examples.

```
solve(u) in Omega by M
  krylov(precond=diagonal)
{
```

Parsing this leads to the modification of a Krylov solver option: it changes the preconditioner from the default none to diagonal. This behaves as expected. Let's now look to a more confusing case:

```
solve(u) in Omega by M
  bicg(epsilon=1E-10, maxiter=1000)
{
```

Parsing will modify *bi-conjugate gradient's* options, epsilon will be 10^{-10} and the *maximum number of iterations* (maxiter) 1000.

⁶the list is extracted from the code automatically to make it up-to-date. The drawback of this is that reading it may be boring, it is to be considered as a reference.

There is no mistake here, but this option will have *no effect* during execution! The reason is simple. The *Krylov solver's* options are not modified and that default linear system solver is *conjugate gradient* (cg). So, *bi-conjugate gradient* will never start and will have no opportunity to read its options.

To have those options used one has to specify the use of bicg:

```
solve(u) in Omega by M
  bicg(epsilon=1E-10, maxiter=1000),
  krylov(type=bicg)
{
```

To conclude with options, consider the code

```
1 solve(u) in Omega by M
2   cg(epsilon=1E-10),
3   cg(epsilon=1E-3,maxiter=15),
4   cg(maxiter=200),
5   {
```

First the epsilon parameter of cg is set to 10^{-10} , at line 2. Then it is change to 10^{-3} and maxiter becomes 15. Finally, at line 4, maxiter is modified to become 200. So *conjugate gradient* will run with $\epsilon = 10^{-3}$ and maxiter=200.

Between the two brackets ({ and }) comes the problem description. We will first focus one *PDE system-like* descriptions and then on *Variational formula-like* descriptions.

PDE and System of PDEs. The general principle is that equations and boundary conditions are defined variable per variable, even in the case of a coupled system. Each new PDE is announced using the `pde(<unknownid>)` structure.

Remark 10 *It helps to rememeber that FreeFEM3D uses this information to construct a variational formulation ⁷. If the informations are not given at the right place, the reconstructed variational formulation may not be the right one, and yet no error message will appear. Moreover, one may need to repeat informations when defining PDE systems, this will be clear in the second example below.*

Here is a simple example for illustration

```
solve (u) in 0 by M
{
  pde (u)
    -div(grad(u)) = 1;
    u = 1 on M;
}
```

⁷This variational formulation is needed by the *finite element* discretization process.

it stands for solving:

$$\begin{cases} -\Delta u = 1 \text{ in } \Omega, \\ u = 1 \text{ on } \partial\Omega \cap \partial M, \\ \frac{\partial u}{\partial n} = 0 \text{ on } \partial\Omega \setminus \partial M. \end{cases}$$

The first remark is that FreeFEM3D and the mathematical syntax are quite close, but that the condition $\frac{\partial u}{\partial n} = 0$ on $\partial\Omega \setminus \partial M$ is *implicit* (uniqueness of the solution assumes that $\text{mes}(\partial\Omega \cap \partial M) > 0$) in FreeFEM3D because the user has forgot to specify what boundary condition to apply on that boundary. For more details on boundary conditions see section 4.5.2, and 4.5.3.

Now look a second example which focuses on the a system description and the *underlying variational problem* as suggested in remark 10.

Lets solve the following problem

$$\begin{cases} -\Delta u - \frac{1}{2}\Delta v = f_1 \text{ in } \Omega, \\ -\frac{1}{2}\Delta u - \Delta v = f_2 \text{ in } \Omega, \\ u = u_0 \text{ on } \Gamma_1, \\ \frac{\partial u}{\partial n} = u_1 \text{ on } \Gamma_2, \\ v = v_0 \text{ on } \Gamma_1, \\ \frac{\partial v}{\partial n} = v_1 \text{ on } \Gamma_2, \end{cases} \quad (4.2)$$

where Ω is the cube $]0, 1[^3$ (for seek of simplicity), $\overline{\Gamma_1 \cup \Gamma_2} = \partial\Omega$, $\Gamma_1 \cap \Gamma_2 = \emptyset$, Γ_2 being the face $x = 0$, and $f_1, f_2, u_0, v_0, u_1, v_1$ given functions such that the problem (4.2) is well posed.

Here comes the associated code in FreeFEM3D:

[example3.txt](#)

```

23 solve(u,v) in Omega by M
24 {
25   pde(u)
26   -div(grad(u))-div(0.5*grad(v))=f1;
27   dnu(u) = u1 on M xmin;
28   dnu(v) = 0.5*v1 on M xmin;
29   u=u0 on M xmax;
30   u=u0 on M ymin;
31   u=u0 on M ymax;
32   u=u0 on M zmin;
33   u=u0 on M zmax;
34
35   pde(v)
36   -div(0.5*grad(u))-div(grad(v))=f2;
37   dnu(u) = 0.5*u1 on M xmin;
38   dnu(v) = v1 on M xmin;
39   v=v0 on M xmax;
40   v=v0 on M ymin;
41   v=v0 on M ymax;
42   v=v0 on M zmin;
43   v=v0 on M zmax;
44 }
```

The lines 28 and 37 are very important! Even if they are redundant they must be provided by the user since the Green formula is not computed by FreeFEM3D, but only a correspondence is made between PDE operators and variational operators.

This comes from the fact that the variational formula associated to (4.2) is

$$\begin{aligned} \int_{\Omega} \nabla u \cdot \nabla w_1 - \int_{\Gamma_2} \nabla u \cdot n w_1 + \frac{1}{2} \int_{\Omega} \nabla v \cdot \nabla w_1 - \boxed{\frac{1}{2} \int_{\Gamma_2} \nabla v \cdot n w_1} \\ + \frac{1}{2} \int_{\Omega} \nabla u \cdot \nabla w_2 - \boxed{\frac{1}{2} \int_{\Gamma_2} \nabla u \cdot n w_2} + \int_{\Omega} \nabla v \cdot \nabla w_2 - \int_{\Gamma_2} \nabla v \cdot n w_2 \\ = \int_{\Omega} f_1 w_1 + \int_{\Omega} f_2 w_2. \end{aligned} \quad (4.3)$$

For given w_1 and w_2 .

The boxed terms are the one which should not be forgotten! Using the information coming from (4.2) boundary conditions, one writes:

$$\begin{cases} \frac{1}{2} \int_{\Gamma_2} \nabla u \cdot n w_2 = \frac{1}{2} \int_{\Gamma_2} u_1 w_2, \text{ and} \\ \frac{1}{2} \int_{\Gamma_2} \nabla v \cdot n w_1 = \frac{1}{2} \int_{\Gamma_2} v_1 w_1. \end{cases}$$

This leads to the final variational formula:

$$\begin{aligned} \int_{\Omega} \nabla u \cdot \nabla w_1 + \frac{1}{2} \int_{\Omega} \nabla v \cdot \nabla w_1 + \frac{1}{2} \int_{\Omega} \nabla u \cdot \nabla w_2 + \int_{\Omega} \nabla v \cdot \nabla w_2 \\ = \int_{\Omega} f_1 w_1 + \int_{\Omega} f_2 w_2 + \int_{\Gamma_2} u_1 w_1 + \boxed{\frac{1}{2} \int_{\Gamma_2} v_1 w_1} + \boxed{\frac{1}{2} \int_{\Gamma_2} u_1 w_2} + \int_{\Gamma_2} v_1 w_2. \end{aligned} \quad (4.4)$$

It would not be an easy task to make FreeFEM3D automagically compute (4.4), which means not forgetting the boxed terms. So those terms have to be provided explicitly by the user following the example. Some of the translations are given bellow.

FreeFEM3D	interpretation	comment
<code>-div(grad(u))</code>	$\int_{\Omega} \nabla u \cdot \nabla w_1$	it is w_1 since we are describing the pde(u)
<code>dnu(u)=u1</code>	$\int_{\Omega} u_1 w_1$	goes to right hand side
<code>dnu(v)=0.5*v1</code>	$\frac{1}{2} \int_{\Omega} v_1 w_1$	<i>it will not be deduced from the pde(v) bloc!</i>

One then understands the logic behind it and can look at the table 4.3 for the complete list of domain operators interpretation.

Variational problem description. Entering a problem with a variational formula is quite different from giving its PDE system. First there is *only* one variational formula (even in the case of systems) and second only the Dirichlet conditions are given outside that formula, since the Neuman and Robin conditions are included in the variational formula.

For example to solve the problem: find u in $H^1(\Omega)$ such that

$$\begin{cases} -\nabla \cdot \mu \nabla u = 0 \text{ in } \Omega, \\ u + \frac{\partial u}{\partial n} = g \text{ on } \Gamma_1, \\ u = u_0 \text{ on } \Gamma_2 = \partial\Omega \setminus \Gamma_1. \end{cases}$$

one sets the variational problem

$$\int_{\Omega} \mu \nabla u \cdot \nabla w - \int_{\partial\Omega} \mu \nabla u \cdot n w = 0 \quad \forall w.$$

Using the fact that

$$\int_{\partial\Omega} \mu \nabla u \cdot n w = \int_{\partial\Omega} \mu (g - \alpha u) w$$

the variational problem is written, find $u \in H^1(\Omega)$ such that $u = u_0$ on Γ_2 and

$$\int_{\Omega} \mu \nabla u \cdot \nabla w + \int_{\partial\Omega} \mu u w = \int_{\partial\Omega} \mu g w \quad \forall w, \quad (4.5)$$

It is this formula (4.5) that must be entered in the FreeFEM3D code

```

13 solve(u) in Omega by M
14 {
15   test(w)
16   int(mu*grad(u)*grad(w)) + int(M xmin)(mu*u*w) = int(M xmin)(mu*g*w);
17   u=uexact on M xmax;
18   u=uexact on M ymin;
19   u=uexact on M ymax;
20   u=uexact on M xmin;
21   u=uexact on M xmax;
22 }
```

Note the presence of the `test(w)` statement. It is here to define a (list of) test functions used in the *bilinear* forms. The *test function variables* only live within the solve bloc⁸.

Then comes the variational formula. As one can see, it is really close to the mathematics, but still needs some explanation:

`int(mu*grad(u)*grad(w))` defines $\int_{\Omega} \mu \nabla u \cdot \nabla w$. The integration domain is implicit;

`int(M xmin)(mu*u*w)` corresponds to $\int_{\Gamma_1} \mu u w$ with Γ_1 being the face $x = 0$ of the domain; and

`int(M xmin)(mu*g*w)` which is $\int_{\Gamma_1} \mu g w$.

The last lines define the Dirichlet boundary conditions. Note that as in section 4.5.3, only Dirichlet conditions are allowed at this level of the problem description.

This example shows what is a “FreeFEM3D variational formula”. Basically, it is an equation made of linear and bilinear terms. A linear form is

$$w \longmapsto l(w)$$

where w must be a *test function*. A bilinear form is

$$(u, w) \longmapsto a(u, w)$$

⁸This is an exception but test function names are not really variables, but tags...

where w must be a test function too and u must be an unknown. All others combination are forbidden!

So the general form of a “FreeFEM3D variational formula” is

$$\sum_j \sum_i a_{ij}(u_i, w_j) = \sum_j l_j(w_j).$$

Where (a_{ij}) is a family of *bilinear forms*, (l_j) is a family of *linear forms*, (u_i) is a family of *unknowns* and (w_j) is a family of *test functions*.

Bilinear and linear forms will be described more precisely in the sections 4.5.4 and 4.5.5.

Lets now reconsider the problem (4.2) and solve it using a variational formula.

The associated variational formula is still given by (4.4). So, the FreeFEM3D code is written immediately by

example5.txt

```

23 solve(u,v) in Omega by M
24   memory(matrix=None)
25 {
26   test(w1,w2)
27   int(grad(u)*grad(w1)) + int(0.5*grad(v)*grad(w1))
28   + int(0.5*grad(u)*grad(w2)) + int(grad(v)*grad(w2))
29   = int(f1*w1) + int(f2*w2)
30     + int(M xmin)(v1*w2 + u1*w1)
31     + int(M xmin)(0.5*v1*w1 + 0.5*u1*w2);
32   u=u0 on M xmax;
33   u=u0 on M ymin;
34   u=u0 on M ymax;
35   u=u0 on M zmin;
36   u=u0 on M zmax;
37   v=v0 on M xmax;
38   v=v0 on M ymin;
39   v=v0 on M ymax;
40   v=v0 on M zmin;
41   v=v0 on M zmax;
42 }
```

This really looks like (4.4)!

4.5.2 PDE system syntax.

The PDE structure uses the syntax

```
[ - ] <pdeoperator> [ { + | - } <pdeoperator> ] ... = <function>;
```

Supported PDE operators are shown on the table 4.3.

Some examples:

```
u - div(grad(u)) = f;
```

FreeFEM3D operator	mathematical	bilinear form ($\forall v$)
$\text{mu}*u$	μu	$\int \mu uv$
$\text{dx}(u)$	$\partial_x u$	$\int \partial_x uv$
$\text{dy}(u)$	$\partial_y u$	$\int \partial_y uv$
$\text{mu}*dz(u)$	$\mu \partial_z u$	$\int \mu \partial_z uv$
$\text{div}(\text{grad}(u))$	$\nabla \cdot \nabla u = \Delta u$	$-\int \nabla u \nabla v$
$\text{dx}(\text{dx}(u)) + \text{dy}(\text{dy}(u)) + \text{dz}(\text{dz}(u))$	$\nabla \cdot \nabla u = \Delta u$	$-\int \partial_x u \partial_x v + \partial_y u \partial_y v + \partial_z u \partial_z v$
$\text{dx}(\text{mu}*dy(u))$	$\partial_x \mu \partial_y u$	$-\int \mu \partial_y u \partial_x v$

Table 4.3: partial differential operators. u is an unknown and mu a function representing μ . The third column shows the bilinear operator that will be used to discretize the partial differential operator, note that in the case of second order operators a Green formula is used. **This means that border term are to be supplied by user on need**, thru boundary conditions.

stands for $u - \nabla \cdot \nabla u = f$.

$$-\text{dx}(\text{dy}(u)) - \text{dy}(\text{dx}(u)) + \text{dz}(u) = f;$$

stands for $-\partial_x \partial_y u - \partial_y \partial_x u + \partial_z u = f$.

4.5.3 Boundary Conditions

To define boundary conditions, one has to use

<condition> on <border>;

The <condition> is defined using the followings:

Dirichlet: $u = g$ is written $u=g$,

Neumann: $\nu \partial_n u = g$ is written $\text{dnu}(u)=g$,

Robin (Fourrier): $\alpha u + \nu \partial_n u = g$ is written $\text{alpha}*u+\text{dnu}(u)=g$.

Remark 11 $\text{dnu}(u)$ denotes the co-normal derivative, the term which arises in the variational form when applying the Green formula to the second order operator. In the case of $-\nabla \cdot \mu \nabla u$ the term is $\mu \nabla u$, coming from

$$-\int_{\Omega} \nabla \cdot \mu \nabla uv = \int_{\Omega} \mu \nabla u \cdot \nabla v + \int_{\partial\Omega} \mu \nabla u \cdot \mathbf{n} v \quad \forall v.$$

To set borders one uses the following syntax:

- on <a,b,c> when the condition is applied on the border of the object having a POV-Ray reference <a,b,c>.
- on M (where M is a *structured mesh*) when the condition is to be applied on the border of M.
- on M <modifier> where <modifier> is one of xmin, xmax, ymin, ymax, zmin or zmax, means that it will be applied to the corresponding face on the *structured mesh* M.

- on S (where S is a *surface mesh*) is used to impose a condition on an already built *surface mesh* (read in a file or previously built).

For example:

$$u = 0 \text{ on } M;$$

$$u = 0 \text{ on } \partial M.$$

$$\text{dnu}(u) = g \text{ on } \langle 1, 0, 0 \rangle;$$

imposes a Neumann condition on the border of objects $\langle 1, 0, 0 \rangle$, the co-normal derivative of u will be equal to g in a weak sense. Similarly

$$u + \text{dnu}(u) = g \text{ on } S;$$

a Robin condition on the surface meshed by S .

4.5.4 Bilinear forms.

Bilinear forms implemented in FreeFEM3D are of the type

$$a(u, w) = \int_{\mathcal{O}} \mathcal{A}(u, w),$$

where \mathcal{O} is an \mathbb{R}^3 domain or an \mathbb{R}^3 surface, and \mathcal{A} is such that

$$\mathcal{A}(u, w) = \sum_i g^i \mathcal{D}_u^i(u) \mathcal{D}_w^i(w),$$

with \mathcal{D}_u^i and \mathcal{D}_w^i being two partial differential operators of order 0 or 1; and u, w two functions. To be a “FreeFEM3D bilinear form”, it is required for u to be an *unknown* and for w to be a *test function*.

The possible choices for operators \mathcal{D}_u and \mathcal{D}_w are given at table 4.4.

FreeFEM3D operator	mathematical meaning
v	v (order 0 operator)
$\text{dx}(v)$	$\partial_x v$
$\text{dy}(v)$	$\partial_y v$
$\text{dz}(v)$	$\partial_z v$
$\text{grad}(v)$	∇v

Table 4.4: partial differential operators. v is an *unknown* or a *test function*.

Note that using variational formula, one can discretize $\int \partial_{x_i} uv$ or $\int u \partial_{x_i} v$ while only $\int \partial_{x_i} uv$ is used in PDEs (see table 4.3).

4.5.5 Linear forms.

In the same way, linear forms implemented in FreeFEM3D are defined as

$$l(w) = \int_{\mathcal{O}} \sum_i \mathcal{L}^i(w).$$

Where \mathcal{O} is an \mathbb{R}^3 domain or an \mathbb{R}^3 surface. \mathcal{L} can be of the following forms:

$$\left| \begin{array}{l} \mathcal{L}(w) = gw, \text{ or} \\ \mathcal{L}(w) = g \nabla f \cdot \nabla w. \end{array} \right.$$

Let us write some examples combining linear and bilinear forms.

```
int(<1,0,0>)(2*u*alpha*v)+int(grad(v)*grad(u))-int(v)=0;
```

is associated to $\int_{\Omega} \nabla u \nabla v + \int_{\Gamma_i} 2\alpha uv = \int_{\Omega} v$, where Γ_i is the border referenced by $\langle 1, 0, 0 \rangle$. By the way, solving

```
int(u*v)=int(f*v*g);
```

for all v , makes u the L^2 projection of fg on the finite element space.

4.5.6 convection operator.

The convection operator

$$\partial_t \varphi + u_i \cdot \partial_{x_i} \varphi, \quad (4.6)$$

can be implemented by using a discrete method of characteristics. To call this operator one uses

```
convect(phi,ux,uy,uz);
```

ϕ is the transported function at the speed (ux, uy, uz) .

One has to note that this function is evaluated *when needed*. This means that if one writes

```
1 function f = convect(phi,ux,uy,uz);
```

f is 'convect(phi,ux,uy,uz)'.

the convect operator adapts to the context:

```
1 function f = convect(phi,ux,uy,uz);
2 femfunction g(M) = f;
3 solve (h) in Omega by M
4 {
5     test(v)
6     int(h*v)=int(f*v);
7 }
```

The g function will be a Q_1 function with values at vertices of the mesh M which are the same as those of f , but to compute h , the values of f will be evaluated at the quadrature vertices.

4.6 Other Instructions.

4.6.1 Input and output.

Built-in. A *built-in* input and output instruction set can be used to read/write files using complex formats. User cannot really change them but can provide options.

Those functions are typically variants of `save` and `read`. Their syntax is

```
save(<format>, <filename>, [<function>,<mesh>[, <filetype>]]);
```

With the following options:

- `<format>` is the data storage *format*, one can refer to the table 4.5 for the list of supported formats,

Format	Identifier
OpenDX	opendx
MEdit (mesh)	medit

Table 4.5: Supported file formats

- `<filename>` is a *string* containing the name of the file stored on the disk,
- the *optional* parameter `<function>` is used to refer to a *function* (usually an *function variable*), if this argument is specified, the given *function* will be stored in the at mesh vertices; if not, the *mesh* will be saved. *Note that only one of the mesh or the function will be saved at once.* This is done to avoid saving the mesh each time a function is stored.
- `<mesh>` is used to define the mesh which will be used to proceed saving.
- The last parameter `<filetype>` is the second *optional* parameter, it is used to define the file type. Possible file types are given on table 4.6.1

Format	Identifier
Unix	unix
MS-DOS/Windows	dos
Machintosh	mac
Binary	binary

Table 4.6: File types

User “defined” This provides more basic stuffs to allow the user to read/write to console or files. The way to use them is similar to *C++* streams so one manipulates low level objects and builds his own format.

The syntax is the following

```
<ostream> [ << <expression> ] ... ;
```

By now the only stream implemented is the ostream cout but streams to files and input stream will be easily introduced. In the same way just strings and doubles can be output. Here comes a simple example.

```
double i = 10;
cout << "i=";
cout << i << "\n";
cout << "----\n";
```

produces the following output:

```
i=10
----
```

4.6.2 Statements.

Syntax for statements in FreeFEM3D follows the rules of C or C++.

Conditional statements. In FreeFEM3D, only the if statement is implemented. Its syntax is

```
if (<boolexp>) { <instruction>; | <bloc> }
[ else { <instruction>; | <bloc> } ]
```

Loops. Standard do-while, while and for structures are implemented:

```
do { <instruction> | <bloc> } while (<boolexp>);
```

```
while (<boolexp>) { <instruction>; | <bloc> }
```

```
for (<instruction>;<boolexp>;<instruction>) { <instruction>; | <bloc> }
```

The following contains a set of examples:

statements.txt

```
1 for (double i=0; i<5; i=i+1)
2   if (i<3)
3     cout << i << " ";
4   else {
5     double j=0;
6     while (j<i) {
7       cout << j-i << " ";
8       do {
9         j=j+1;
10      } while (0>1);
11    }
12  }
13 cout << "\n";
```

It produces the following output

```
0 1 2 -3 -2 -1 -4 -3 -2 -1
```


Chapter 5

Examples

5.1 A Simple Example: the Poisson Problem in a Cube.

Find u such that

$$\begin{cases} -\Delta u = f \text{ in } \Omega, \\ u|_{\Gamma} = 0, \end{cases} \quad (5.1)$$

where $f \in L^2(\Omega)$, and $\Gamma = \partial\Omega$. Lets assume that $\Omega = (-1, 1)^3$, and choose $f = 1$.

Here the domain is a cube so there will be no informations coming from the POV-Rayfile. The mesh is built by

```
mesh M = structured((-1,1,-1),(1,-1,1),(10,10,10));
```

We recall that the mesh M is built in a box specified by its two opposite corners $(-1, 1, -1)$ and $(1, -1, 1)$ and $(10, 10, 10)$ specifies the number of discretization points on each edge.

For more clarity one may prefer the following notations:

```
1 vector n = (10, 10, 10);
2 vertex a = (-1, 1,-1);
3 vertex b = ( 1,-1, 1);
4 mesh M = structured(n,a,b);
```

Since the geometry is very simple here one has to create an empty *scene*. This POV-Ray scene will be describe by an empty file: "empty.pov". The *domain* Ω will be declared by

```
6 scene S = pov("void.pov"); // the pov-ray file for the geometry
7 domain O = domain(S);
```

The PDE is specified by

```
9 solve(u) in O by M
10 {
11     pde(u)
12     - div(grad(u)) = 1;
13     u = 0 on M;
14 };
15 save(opendx,"u.dat",u,M);
```

The last line is used to save the data in the file "u.dat".

Index

domain, 12

double, 12

function, 12

mesh, 12

method, 3

scene, 12

vector, 12

vertex, 12